

Catoids, Categories, Groupoids

Georg Struth

March 17, 2025

Abstract

This AFP entry formalises catoids, which are generalisations of single-set categories, and groupoids. More specifically, in catoids, the partial composition of arrows in a category is generalised to a multi-operation, which sends pairs of elements to sets of elements, and the definedness condition of arrow composition – two arrows can be composed if and only the target of the first matches the source of the second – is relaxed. Beyond a library of basic laws for catoids, single-set categories and groupoids, I formalise the facts that every catoid can be lifted to a modal powerset quantale, that every groupoid can be lifted to a Dedekind quantale and to power set relation algebras, a special case of a famous result of Jónsson and Tarski. Finally, I show that single-set categories are equivalent to a standard axiomatisation of categories based on a set of objects and a set of arrows, and compare catoids with related structures such as multimonoid and relational monoids (monoids in the monoidal category Rel).

Contents

1	Introductory Remarks	2
2	Catoids	3
2.1	Multimagmas	3
2.2	Multisemigroups	4
2.3	st-Multimagmas	4
2.4	Catoids	8
2.5	Locality	10
2.6	From partial magmas to single-set categories.	11
2.7	Morphisms of multimagmas and lr-multimagmas	12
2.8	Relationship with categories	13
2.9	A Mac Lane style variant	14
2.10	Product of catoids	15

3	Groupoids	15
3.1	st-Multigroupoids	15
3.2	Groupoids	17
3.3	Axioms of relation algebra	19
4	Lifting catoids to modal powerset quantales	20
5	Lifting groupoids to powerset Dedekind quantales and powerset relation algebras	22
6	Multimonoids	23
6.1	Unital multimagnas	23
6.2	Multimonoids	25
6.3	Multimonoids and catoids	26
6.4	From multimonoids to categories	27
6.5	Multimonoids and relational monoids	28

1 Introductory Remarks

These Isabelle theories formalise results on catoids from [4, 2, 6] and groupoids from [3]. Catoids generalise single-set categories, as they can be found in Chapter XII of Mac Lane’s book [8]. One particular result, namely that catoids can be lifted to (power set) relation algebras, is due to Jónsson and Tarski [7].

A wide-ranging formalisation of category theory based on single-set categories formalised as locales can already be found in the AFP [9]. The present type-class-based alternative might lend itself to a similar programme.

The multioperation $X \times X \rightarrow \mathcal{P}X$ in the definition of catoids is obviously isomorphic to a ternary relation $X \rightarrow X \rightarrow X \rightarrow 2$. Simple mathematical components for relational monoids, which are isomorphic (as categories with suitable morphisms) to catoids, can already be found in the AFP [5]. At this stage, I do not integrate the two components. They use different formalisations of quantales with Isabelle, which remain to be consolidated.

Catoids and groupoids admit many models. Those of catoids range from shuffle monoids and generalised effect algebras to base algebras of incidence and matrix algebras [6], whereas groupoids are so ubiquitous in mathematics that some mathematicians have argued for interchanging their names with groups, see [1] for a brief history, which goes decades beyond that of category theory.

The mathematical components in this AFP entry are also stepping stones towards the formalisation of (ω, p) -catoids, strict (ω, p) -categories and (ω, p) -quantales. Components for these structures will feature in a separate AFP

entry. They contribute to a larger programme on the formalisation of higher rewriting techniques with proof assistants.

I am grateful for a fellowship at the Collegium de Lyon, Institute for Advanced Study, where this formalisation work has been done.

2 Catoids

```
theory Catoid
  imports Main
```

```
begin
```

2.1 Multimagnas

Multimagnas are sets equipped with multioperations. Multioperations are isomorphic to ternary relations.

```
class multimagma =
  fixes mcomp :: 'a ⇒ 'a ⇒ 'a set (infixl <⊙> 70)
```

```
begin
```

I introduce notation for the domain of definition of the multioperation.

```
abbreviation Δ x y ≡ (x ⊙ y ≠ {})
```

I extend the multioperation to powersets

```
definition conv :: 'a set ⇒ 'a set ⇒ 'a set (infixl <★> 70) where
  X ★ Y = (⋃ x ∈ X. ⋃ y ∈ Y. x ⊙ y)
```

```
lemma conv-exp: X ★ Y = {z. ∃ x y. z ∈ x ⊙ y ∧ x ∈ X ∧ y ∈ Y}
  <proof>
```

```
lemma conv-exp2: (z ∈ X ★ Y) = (∃ x y. z ∈ x ⊙ y ∧ x ∈ X ∧ y ∈ Y)
  <proof>
```

```
lemma conv-distl: X ★ ⋃ Y = (⋃ Y ∈ Y. X ★ Y)
  <proof>
```

```
lemma conv-distr: ⋃ X ★ Y = (⋃ X ∈ X. X ★ Y)
  <proof>
```

```
lemma conv-distl-small: X ★ (Y ∪ Z) = X ★ Y ∪ X ★ Z
  <proof>
```

```
lemma conv-distr-small: (X ∪ Y) ★ Z = X ★ Z ∪ Y ★ Z
  <proof>
```

lemma *conv-isol*: $X \subseteq Y \implies Z \star X \subseteq Z \star Y$
 ⟨*proof*⟩

lemma *conv-isor*: $X \subseteq Y \implies X \star Z \subseteq Y \star Z$
 ⟨*proof*⟩

lemma *conv-atom* [*simp*]: $\{x\} \star \{y\} = x \odot y$
 ⟨*proof*⟩

end

2.2 Multisemigroups

Sultisemigroups are associative multimagnas.

class *multisemigroup* = *multimagma* +
assumes *assoc*: $(\bigcup v \in y \odot z. x \odot v) = (\bigcup v \in x \odot y. v \odot z)$

begin

lemma *assoc-exp*: $(\exists v. w \in x \odot v \wedge v \in y \odot z) = (\exists v. v \in x \odot y \wedge w \in v \odot z)$
 ⟨*proof*⟩

lemma *assoc-var*: $\{x\} \star (y \odot z) = (x \odot y) \star \{z\}$
 ⟨*proof*⟩

Associativity extends to powersets.

lemma *conv-assoc*: $X \star (Y \star Z) = (X \star Y) \star Z$
 ⟨*proof*⟩

end

2.3 st-Multimagnas

I equip multimagnas with source and target maps.

class *st-op* =
fixes *src* :: 'a \Rightarrow 'a ($\langle\sigma\rangle$)
and *tgt* :: 'a \Rightarrow 'a ($\langle\tau\rangle$)

class *st-multimagma* = *multimagma* + *st-op* +
assumes *Dst*: $x \odot y \neq \{\}$ $\implies \tau x = \sigma y$
and *s-absorb* [*simp*]: $\sigma x \odot x = \{x\}$
and *t-absorb* [*simp*]: $x \odot \tau x = \{x\}$

The following sublocale proof sets up opposition/duality.

sublocale *st-multimagma* \subseteq *stopp*: *st-multimagma* $\lambda x y. y \odot x$ *tgt src*
rewrites *stopp.conv* $X Y = Y \star X$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *ts-compat* [*simp*]: $\tau (\sigma x) = \sigma x$
<proof>

lemma (in *st-multimagma*) *ss-idem* [*simp*]: $\sigma (\sigma x) = \sigma x$
<proof>

lemma (in *st-multimagma*) *st-fix*: $(\tau x = x) = (\sigma x = x)$
<proof>

lemma (in *st-multimagma*) *st-eq1*: $\sigma x = x \implies \sigma x = \tau x$
<proof>

lemma (in *st-multimagma*) *st-eq2*: $\tau x = x \implies \sigma x = \tau x$
<proof>

I extend source and target operations to powersets by taking images.

abbreviation (in *st-op*) *Src* :: 'a set \Rightarrow 'a set **where**
Src \equiv image σ

abbreviation (in *st-op*) *Tgt* :: 'a set \Rightarrow 'a set **where**
Tgt \equiv image τ

Fixpoints of source and target maps model source and target elements.
These correspond to units.

abbreviation (in *st-op*) *sfix* :: 'a set **where**
sfix \equiv {x. $\sigma x = x$ }

abbreviation (in *st-op*) *tfix* :: 'a set **where**
tfix \equiv {x. $\tau x = x$ }

lemma (in *st-multimagma*) *st-mm-rfix* [*simp*]: *tfix* = *stopp.sfix*
<proof>

lemma (in *st-multimagma*) *st-fix-set*: {x. $\sigma x = x$ } = {x. $\tau x = x$ }
<proof>

lemma (in *st-multimagma*) *stfix-set*: *sfix* = *tfix*
<proof>

lemma (in *st-multimagma*) *sfix-im*: *sfix* = *Src UNIV*
<proof>

lemma (in *st-multimagma*) *tfix-im*: *tfix* = *Tgt UNIV*
<proof>

lemma (in *st-multimagma*) *ST-im*: *Src UNIV* = *Tgt UNIV*
<proof>

Source and target elements are "orthogonal" idempotents.

lemma (in *st-multimagma*) *s-idem* [*simp*]: $\sigma x \odot \sigma x = \{\sigma x\}$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *s-ortho*:
 $\Delta (\sigma x) (\sigma y) \implies \sigma x = \sigma y$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *s-ortho-iff*: $\Delta (\sigma x) (\sigma y) = (\sigma x = \sigma y)$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *st-ortho-iff*: $\Delta (\sigma x) (\tau y) = (\sigma x = \tau y)$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *s-ortho-id*: $(\sigma x) \odot (\sigma y) = (\text{if } (\sigma x = \sigma y) \text{ then } \{\sigma x\} \text{ else } \{\})$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *s-absorb-var*: $(\sigma y \neq \sigma x) = (\sigma y \odot x = \{\})$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *s-absorb-var2*: $(\sigma y = \sigma x) = (\sigma y \odot x \neq \{\})$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *s-absorb-var3*: $(\sigma y = \sigma x) = \Delta (\sigma x) y$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *s-assoc*: $\{\sigma x\} \star (\sigma y \odot z) = (\sigma x \odot \sigma y) \star \{z\}$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *sfix-absorb-var* [*simp*]: $(\bigcup e \in \text{sfix}. e \odot x) = \{x\}$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *tfix-absorb-var*: $(\bigcup e \in \text{tfix}. x \odot e) = \{x\}$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *st-comm*: $\tau x \odot \sigma y = \sigma y \odot \tau x$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *s-weak-twisted*: $(\bigcup u \in x \odot y. \sigma u \odot x) \subseteq x \odot \sigma y$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *s-comm*: $\sigma x \odot \sigma y = \sigma y \odot \sigma x$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *s-export* [*simp*]: $\text{Src } (\sigma x \odot y) = \sigma x \odot \sigma y$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *st-prop*: $(\tau x = \sigma y) = \Delta (\tau x) (\sigma y)$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *weak-local-var*: $\tau x \odot \sigma y = \{\}$ $\implies x \odot y = \{\}$
 ⟨*proof*⟩

The following facts hold by duality.

lemma (in *st-multimagma*) *st-compat*: $\sigma (\tau x) = \tau x$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *tt-idem*: $\tau (\tau x) = \tau x$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *t-idem*: $\tau x \odot \tau x = \{\tau x\}$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *t-weak-twisted*: $(\bigcup u \in y \odot x. x \odot \tau u) \subseteq \tau y \odot x$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *t-comm*: $\tau x \odot \tau y = \tau y \odot \tau x$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *t-export*: *image* $\tau (x \odot \tau y) = \tau x \odot \tau y$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *tt-comp-prop*: $\Delta (\tau x) (\tau y) = (\tau x = \tau y)$
 ⟨*proof*⟩

The set of all sources (and targets) are units at powerset level.

lemma (in *st-multimagma*) *conv-uns [simp]*: $\text{sfix} \star X = X$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *conv-unt*: $X \star \text{tfix} = X$
 ⟨*proof*⟩

I prove laws of modal powerset quantales.

lemma (in *st-multimagma*) *Src-exp*: $\text{Src } X = \{\sigma x \mid x. x \in X\}$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *ST-compatible [simp]*: $\text{Src } (\text{Tgt } X) = \text{Tgt } X$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *TS-compatible*: $\text{Tgt } (\text{Src } X) = \text{Src } X$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *Src-absorp [simp]*: $\text{Src } X \star X = X$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *Tgt-absorp*: $X \star \text{Tgt } X = X$
 ⟨*proof*⟩

lemma (in *st-multimagma*) *Src-Sup-pres*: $\text{Src} (\bigcup \mathcal{X}) = (\bigcup X \in \mathcal{X}. \text{Src } X)$
 ⟨proof⟩

lemma (in *st-multimagma*) *Tgt-Sup-pres*: $\text{Tgt} (\bigcup \mathcal{X}) = (\bigcup X \in \mathcal{X}. \text{Tgt } X)$
 ⟨proof⟩

lemma (in *st-multimagma*) *ST-comm*: $\text{Src } X \star \text{Tgt } Y = \text{Tgt } Y \star \text{Src } X$
 ⟨proof⟩

lemma (in *st-multimagma*) *Src-comm*: $\text{Src } X \star \text{Src } Y = \text{Src } Y \star \text{Src } X$
 ⟨proof⟩

lemma (in *st-multimagma*) *Tgt-comm*: $\text{Tgt } X \star \text{Tgt } Y = \text{Tgt } Y \star \text{Tgt } X$
 ⟨proof⟩

lemma (in *st-multimagma*) *Src-subid*: $\text{Src } X \subseteq \text{sfix}$
 ⟨proof⟩

lemma (in *st-multimagma*) *Tgt-subid*: $\text{Tgt } X \subseteq \text{tfix}$
 ⟨proof⟩

lemma (in *st-multimagma*) *Src-export [simp]*: $\text{Src} (\text{Src } X \star Y) = \text{Src } X \star \text{Src } Y$
 ⟨proof⟩

lemma (in *st-multimagma*) *Tgt-export [simp]*: $\text{Tgt} (X \star \text{Tgt } Y) = \text{Tgt } X \star \text{Tgt } Y$
 ⟨proof⟩

Locality implies st-locality, which is the composition pattern of categories.

lemma (in *st-multimagma*) *locality*:
assumes *src-local*: $\text{Src} (x \odot \sigma y) \subseteq \text{Src} (x \odot y)$
and *tgt-local*: $\text{Tgt} (\tau x \odot y) \subseteq \text{Tgt} (x \odot y)$
shows $\Delta x y = (\tau x = \sigma y)$
 ⟨proof⟩

2.4 Catoids

class *catoid* = *st-multimagma* + *multisemigroup*

sublocale *catoid* \subseteq *ts-msg*: *catoid* $\lambda x y. y \odot x$ *tgt src*
 ⟨proof⟩

lemma (in *catoid*) *src-comp-aux*: $v \in x \odot y \implies \sigma v = \sigma x$
 ⟨proof⟩

lemma (in *catoid*) *src-comp*: $\text{Src} (x \odot y) \subseteq \{\sigma x\}$
 ⟨proof⟩

lemma (in *catoid*) *src-comp-cond*: $(\Delta x y) \implies \text{Src} (x \odot y) = \{\sigma x\}$
 ⟨proof⟩

lemma (in *catoid*) *tgt-comp-aux*: $v \in x \odot y \implies \tau v = \tau y$
 ⟨*proof*⟩

lemma (in *catoid*) *tgt-comp*: $Tgt(x \odot y) \subseteq \{\tau y\}$
 ⟨*proof*⟩

lemma (in *catoid*) *tgt-comp-cond*: $\Delta x y \implies Tgt(x \odot y) = \{\tau y\}$
 ⟨*proof*⟩

lemma (in *catoid*) *src-weak-local*: $Src(x \odot y) \subseteq Src(x \odot \sigma y)$
 ⟨*proof*⟩

lemma (in *catoid*) *src-local-cond*:
 $\Delta x y \implies Src(x \odot y) = Src(x \odot \sigma y)$
 ⟨*proof*⟩

lemma (in *catoid*) *tgt-weak-local*: $Tgt(x \odot y) \subseteq Tgt(\tau x \odot y)$
 ⟨*proof*⟩

lemma (in *catoid*) *tgt-local-cond*:
 $\Delta x y \implies Tgt(x \odot y) = Tgt(\tau x \odot y)$
 ⟨*proof*⟩

lemma (in *catoid*) *src-twisted-aux*:
 $u \in x \odot y \implies (x \odot \sigma y = \sigma u \odot x)$
 ⟨*proof*⟩

lemma (in *catoid*) *src-twisted-cond*:
 $\Delta x y \implies x \odot \sigma y = \bigcup \{\sigma u \odot x \mid u. u \in x \odot y\}$
 ⟨*proof*⟩

lemma (in *catoid*) *tgt-twisted-aux*:
 $u \in x \odot y \implies (\tau x \odot y = y \odot \tau u)$
 ⟨*proof*⟩

lemma (in *catoid*) *tgt-twisted-cond*:
 $\Delta x y \implies \tau x \odot y = \bigcup \{y \odot \tau u \mid u. u \in x \odot y\}$
 ⟨*proof*⟩

lemma (in *catoid*) *src-funct*:
 $x \in y \odot z \implies x' \in y \odot z \implies \sigma x = \sigma x'$
 ⟨*proof*⟩

lemma (in *catoid*) *st-local-iff*:
 $(\forall x y. \Delta x y = (\tau x = \sigma y)) = (\forall v x y z. v \in x \odot y \implies \Delta y z \implies \Delta v z)$
 ⟨*proof*⟩

Again one can lift to properties of modal semirings and quantales.

lemma (in *catoid*) *Src-weak-local*: $\text{Src } (X \star Y) \subseteq \text{Src } (X \star \text{Src } Y)$
 ⟨*proof*⟩

lemma (in *catoid*) *Tgt-weak-local*: $\text{Tgt } (X \star Y) \subseteq \text{Tgt } (\text{Tgt } X \star Y)$
 ⟨*proof*⟩

st-Locality implies locality.

lemma (in *catoid*) *st-locality-l-locality*:
assumes $\Delta x y = (\tau x = \sigma y)$
shows $\text{Src } (x \odot y) = \text{Src } (x \odot \sigma y)$
 ⟨*proof*⟩

lemma (in *catoid*) *st-locality-r-locality*:
assumes *lr-locality*: $\Delta x y = (\tau x = \sigma y)$
shows $\text{Tgt } (x \odot y) = \text{Tgt } (\tau x \odot y)$
 ⟨*proof*⟩

lemma (in *catoid*) *st-locality-locality*:
 $(\text{Src } (x \odot y) = \text{Src } (x \odot \sigma y) \wedge \text{Tgt } (x \odot y) = \text{Tgt } (\tau x \odot y)) = (\Delta x y = (\tau x = \sigma y))$
 ⟨*proof*⟩

2.5 Locality

For st-multimagmas there are different notions of locality. I do not develop this in detail.

class *local-catoid* = *catoid* +
assumes *src-local*: $\text{Src } (x \odot \sigma y) \subseteq \text{Src } (x \odot y)$
and *tgt-local*: $\text{Tgt } (\tau x \odot y) \subseteq \text{Tgt } (x \odot y)$

sublocale *local-catoid* \subseteq *sts-msg*: *local-catoid* $\lambda x y. y \odot x$ *tgt src*
 ⟨*proof*⟩

lemma (in *local-catoid*) *src-local-eq* [*simp*]: $\text{Src } (x \odot \sigma y) = \text{Src } (x \odot y)$
 ⟨*proof*⟩

lemma (in *local-catoid*) *tgt-local-eq*: $\text{Tgt } (\tau x \odot y) = \text{Tgt } (x \odot y)$
 ⟨*proof*⟩

lemma (in *local-catoid*) *src-twisted*: $x \odot \sigma y = (\bigcup u \in x \odot y. \sigma u \odot x)$
 ⟨*proof*⟩

lemma (in *local-catoid*) *tgt-twisted*: $\tau x \odot y = (\bigcup u \in x \odot y. y \odot \tau u)$
 ⟨*proof*⟩

lemma (in *local-catoid*) *local-var*: $\Delta x y \implies \Delta (\tau x) (\sigma y)$
 ⟨*proof*⟩

lemma (in *local-catoid*) *local-var-eq* [simp]: $\Delta (\tau x) (\sigma y) = \Delta x y$
 ⟨proof⟩

I lift locality to powersets.

lemma (in *local-catoid*) *Src-local* [simp]: $\text{Src } (X \star \text{Src } Y) = \text{Src } (X \star Y)$
 ⟨proof⟩

lemma (in *local-catoid*) *Tgt-local* [simp]: $\text{Tgt } (\text{Tgt } X \star Y) = \text{Tgt } (X \star Y)$
 ⟨proof⟩

lemma (in *local-catoid*) *st-local*: $\Delta x y = (\tau x = \sigma y)$
 ⟨proof⟩

2.6 From partial magmas to single-set categories.

class *functional-magma* = *multimagma* +
assumes *functionality*: $x \in y \odot z \implies x' \in y \odot z \implies x = x'$

begin

Functional magmas could also be called partial magmas. The multioperation corresponds to a partial operation.

lemma *partial-card*: $\text{card } (x \odot y) \leq 1$
 ⟨proof⟩

lemma *fun-in-sgl*: $(x \in y \odot z) = (\{x\} = y \odot z)$
 ⟨proof⟩

I introduce a partial operation.

definition *pcomp* :: $'a \Rightarrow 'a \Rightarrow 'a$ (**infixl** $\langle \otimes \rangle$ 70) **where**
 $x \otimes y = (\text{THE } z. z \in x \odot y)$

lemma *functionality-var*: $\Delta x y \implies (\exists! z. z \in x \odot y)$
 ⟨proof⟩

lemma *functionality-lem*: $(\exists! z. z \in x \odot y) \vee (x \odot y = \{\})$
 ⟨proof⟩

lemma *functionality-lem-var*: $\Delta x y = (\exists z. \{z\} = x \odot y)$
 ⟨proof⟩

lemma *pcomp-def-var*: $(\Delta x y \wedge x \otimes y = z) = (z \in x \odot y)$
 ⟨proof⟩

lemma *pcomp-def-var2*: $\Delta x y \implies ((x \otimes y = z) = (z \in x \odot y))$
 ⟨proof⟩

lemma *pcomp-def-var3*: $\Delta x y \implies ((x \otimes y = z) = (\{z\} = x \odot y))$
 ⟨proof⟩

end

class *functional-st-magma* = *functional-magma* + *st-multimagma*

class *functional-semigroup* = *functional-magma* + *multisemigroup*

begin

lemma *pcomp-assoc-defined*: $(\Delta u v \wedge \Delta (u \otimes v) w) = (\Delta u (v \otimes w) \wedge \Delta v w)$
<proof>

lemma *pcomp-assoc*: $\Delta x y \wedge \Delta (x \otimes y) z \implies (x \otimes y) \otimes z = x \otimes (y \otimes z)$
<proof>

end

class *functional-catoid* = *functional-semigroup* + *catoid*

Finally, here comes the definition of single-set categories as in Chapter 12 of Mac Lane's book, but with partiality of arrow composition modelled using a multioperation, or a partial operation based on it.

class *single-set-category* = *functional-catoid* + *local-catoid*

begin

lemma *st-assoc*: $\tau x = \sigma y \implies \tau y = \sigma z \implies (x \otimes y) \otimes z = x \otimes (y \otimes z)$
<proof>

end

2.7 Morphisms of multimagnas and lr-multimagnas

In the context of single-set categories, these morphisms are functors. Bounded morphisms are functional bisimulations. They are known as zig-zag morphisms or p-morphism in modal and substructural logics.

definition *mm-morphism* :: (*'a::multimagma* \Rightarrow *'b::multimagma*) \Rightarrow *bool* **where**
mm-morphism *f* = $(\forall x y. \text{image } f (x \odot y) \subseteq f x \odot f y)$

definition *bounded-mm-morphism* :: (*'a::multimagma* \Rightarrow *'b::multimagma*) \Rightarrow *bool*
where

bounded-mm-morphism *f* = $(\text{mm-morphism } f \wedge (\forall x u v. f x \in u \odot v \longrightarrow (\exists y z. u = f y \wedge v = f z \wedge x \in y \odot z)))$

definition *st-mm-morphism* :: (*'a::st-multimagma* \Rightarrow *'b::st-multimagma*) \Rightarrow *bool*
where

st-mm-morphism *f* = $(\text{mm-morphism } f \wedge f \circ \sigma = \sigma \circ f \wedge f \circ \tau = \tau \circ f)$

definition *bounded-st-mm-morphism* :: ('a::st-multimagma \Rightarrow 'b::st-multimagma) \Rightarrow bool **where**

bounded-st-mm-morphism f = (bounded-mm-morphism f \wedge st-mm-morphism f)

2.8 Relationship with categories

Next I add a standard definition of a category following Moerdijk and Mac Lane's book and, for good measure, show that categories form single set categories and vice versa.

locale *category* =

fixes *id* :: 'objects \Rightarrow 'arrows

and *dom* :: 'arrows \Rightarrow 'objects

and *cod* :: 'arrows \Rightarrow 'objects

and *comp* :: 'arrows \Rightarrow 'arrows \Rightarrow 'arrows (**infixl** $\langle \cdot \rangle$ 70)

assumes *dom-id* [*simp*]: *dom* (id X) = X

and *cod-id* [*simp*]: *cod* (id X) = X

and *id-dom* [*simp*]: *id* (dom f) \cdot f = f

and *id-cod* [*simp*]: f \cdot *id* (cod f) = f

and *dom-loc* [*simp*]: *cod* f = dom g \Longrightarrow dom (f \cdot g) = dom f

and *cod-loc* [*simp*]: *cod* f = dom g \Longrightarrow *cod* (f \cdot g) = *cod* g

and *assoc*: *cod* f = dom g \Longrightarrow *cod* g = dom h \Longrightarrow (f \cdot g) \cdot h = f \cdot (g \cdot h)

begin

lemma *cod f = dom g \Longrightarrow dom (f \cdot g) = dom (f \cdot id (dom g))*

\langle proof \rangle

abbreviation *LL* f \equiv *id* (dom f)

abbreviation *RR* f \equiv *id* (cod f)

abbreviation *Comp* \equiv λ f g. (if *RR* f = *LL* g then {f \cdot g} else {})

end

typedef (**overloaded**) 'a::single-set-category *st-objects* = {x::'a::single-set-category.

σ x = x}

\langle proof \rangle

setup-lifting *type-definition-st-objects*

lemma *Sfix-coerce* [*simp*]: *Abs-st-objects* (σ (*Rep-st-objects* X)) = X

\langle proof \rangle

lemma *Rfix-coerce* [*simp*]: *Abs-st-objects* (τ (*Rep-st-objects* X)) = X

\langle proof \rangle

sublocale *single-set-category* \subseteq *sccatcat*: *category Rep-st-objects Abs-st-objects* \circ σ *Abs-st-objects* \circ τ (\otimes)

<proof>

sublocale *category* \subseteq *catlrm: st-multimagma Comp LL RR*
<proof>

sublocale *category* \subseteq *catsscat: single-set-category Comp LL RR*
<proof>

2.9 A Mac Lane style variant

Next I present an axiomatisation of single-set categories that follows Mac Lane's axioms in Chapter I of his textbook more closely, but still uses a multioperation for arrow composition.

```
class mlss-cat = functional-magma +  
  fixes l0 :: 'a  $\Rightarrow$  'a  
  fixes r0 :: 'a  $\Rightarrow$  'a  
  assumes comp0-def: (x  $\odot$  y  $\neq$  {}) = (r0 x = l0 y)  
  assumes r0l0 [simp]: r0 (l0 x) = l0 x  
  assumes l0r0 [simp]: l0 (r0 x) = r0 x  
  assumes l0-absorb [simp]: l0 x  $\otimes$  x = x  
  assumes r0-absorb [simp]: x  $\otimes$  r0 x = x  
  assumes assoc-defined: (u  $\odot$  v  $\neq$  {})  $\wedge$  (u  $\otimes$  v)  $\odot$  w  $\neq$  {} = (u  $\odot$  (v  $\otimes$  w)  $\neq$  {})  
  assumes comp0-assoc: r0 x = l0 y  $\implies$  r0 y = l0 z  $\implies$  x  $\otimes$  (y  $\otimes$  z) = (x  $\otimes$  y)  $\otimes$  z  
  assumes locall-var: r0 x = l0 y  $\implies$  l0 (x  $\otimes$  y) = l0 x  
  assumes localr-var: r0 x = l0 y  $\implies$  r0 (x  $\otimes$  y) = r0 y
```

begin

lemma *ml-locall* [*simp*]: l0 (x \otimes l0 y) = l0 (x \otimes y)
<proof>

lemma *ml-localr* [*simp*]: r0 (r0 x \otimes y) = r0 (x \otimes y)
<proof>

lemma *ml-locall-im* [*simp*]: image l0 (x \odot l0 y) = image l0 (x \odot y)
<proof>

lemma *ml-localr-im* [*simp*]: image r0 (r0 x \odot y) = image r0 (x \odot y)
<proof>

end

sublocale *single-set-category* \subseteq *sscatml: mlss-cat (\odot) σ τ*
<proof>

sublocale *mlss-cat* \subseteq *mlsscat: single-set-category (\odot) l0 r0*
<proof>

2.10 Product of catoids

Finally I formalise products of categories as an exercise.

```
instantiation prod :: (catoid, catoid) catoid
begin
```

```
definition src-prod x = ( $\sigma$  (fst x),  $\sigma$  (snd x))
for x :: 'a × 'b
```

```
definition tgt-prod x = ( $\tau$  (fst x),  $\tau$  (snd x))
for x :: 'a × 'b
```

```
definition mcomp-prod x y = {(u,v) | u v. u ∈ fst x ⊙ fst y ∧ v ∈ snd x ⊙ snd y}
for x y :: 'a × 'b
```

```
instance
⟨proof⟩
```

```
end
```

```
instantiation prod :: (single-set-category, single-set-category) single-set-category
begin
```

```
instance
⟨proof⟩
```

```
end
```

```
end
```

3 Groupoids

```
theory Groupoid
imports Catoid
```

```
begin
```

3.1 st-Multigroupoids

I define multigroupoids, extending the standard definition. I equip catoids with an operation of inversion.

```
class inv-op = fixes inv :: 'a ⇒ 'a
```

```
class st-multigroupoid = catoid + inv-op +
assumes invl:  $\sigma$  x ∈ x ⊙ inv x
and invr:  $\tau$  x ∈ inv x ⊙ x
```

sublocale $st\text{-multigroupoid} \subseteq st\text{-mgpd}$: $st\text{-multigroupoid} \ \lambda x \ y. \ y \odot x \ \text{tgt} \ \text{src} \ \text{inv}$
 $\langle proof \rangle$

Every multigroupoid is local.

lemma (in $st\text{-multigroupoid}$) $st\text{-mgpd}\text{-local}$:
assumes $\tau \ x = \sigma \ y$
shows $\Delta \ x \ y$
 $\langle proof \rangle$

sublocale $st\text{-multigroupoid} \subseteq stmgpd$: $local\text{-catoid} \ (\odot) \ \text{src} \ \text{tgt}$
 $\langle proof \rangle$

lemma (in $st\text{-multigroupoid}$) $\text{tgt}\text{-inv} \ [simp]$: $\tau \ (\text{inv} \ x) = \sigma \ x$
 $\langle proof \rangle$

lemma (in $st\text{-multigroupoid}$) $\text{src}\text{-inv}$: $\sigma \ (\text{inv} \ x) = \tau \ x$
 $\langle proof \rangle$

The following lemma is from Theorem 5.2 of Jónsson and Tarski's Boolean Algebras with Operators II article.

lemma (in $st\text{-multigroupoid}$) $ba03$:
assumes $x \odot y = \{\sigma \ x\}$
shows $\text{inv} \ x = y$
 $\langle proof \rangle$

lemma (in $st\text{-multigroupoid}$) $\text{inv}\text{-s} \ [simp]$: $\text{inv} \ (\sigma \ x) = \sigma \ x$
 $\langle proof \rangle$

lemma (in $st\text{-multigroupoid}$) $\text{src}\text{funct}\text{-inv}$:
 $\sigma \ x \in x \odot \text{inv} \ x \implies \sigma \ y \in x \odot \text{inv} \ x \implies \sigma \ x = \sigma \ y$
 $\langle proof \rangle$

lemma (in $st\text{-multigroupoid}$) $\text{tgt}\text{funct}\text{-inv}$:
 $\tau \ x \in \text{inv} \ x \odot x \implies \tau \ y \in \text{inv} \ x \odot x \implies \tau \ x = \tau \ y$
 $\langle proof \rangle$

As for catoids, I prove quantalic properties, lifting to powersets.

abbreviation (in $st\text{-multigroupoid}$) $\text{Inv} :: 'a \ \text{set} \Rightarrow 'a \ \text{set}$ **where**
 $\text{Inv} \equiv \text{image} \ \text{inv}$

lemma (in $st\text{-multigroupoid}$) $\text{Inv}\text{-exp}$: $\text{Inv} \ X = \{\text{inv} \ x \mid x. \ x \in X\}$
 $\langle proof \rangle$

lemma (in $st\text{-multigroupoid}$) $\text{Inv}\text{-un}$: $\text{Inv} \ (X \cup Y) = \text{Inv} \ X \cup \text{Inv} \ Y$
 $\langle proof \rangle$

lemma (in $st\text{-multigroupoid}$) $\text{Inv}\text{-Un}$: $\text{Inv} \ (\bigcup \mathcal{X}) = \bigcup \{X \in \mathcal{X}. \ \text{Inv} \ X\}$
 $\langle proof \rangle$

lemma (in *st-multigroupoid*) *Invl*: $\text{Src } X \subseteq X \star \text{Inv } X$
 ⟨*proof*⟩

lemma (in *st-multigroupoid*) *Invr*: $\text{Tgt } X \subseteq \text{Inv } X \star X$
 ⟨*proof*⟩

lemma (in *st-multigroupoid*) *Inv-strong-gelfand*: $X \subseteq X \star \text{Inv } X \star X$
 ⟨*proof*⟩

At powerset level, one can define domain and codomain operations explicitly as in relation algebras.

lemma (in *st-multigroupoid*) *dom-def*: $\text{Src } X = \text{sfix} \cap (X \star \text{Inv } X)$
 ⟨*proof*⟩

lemma (in *st-multigroupoid*) *cod-def*: $\text{Tgt } X = \text{sfix} \cap (\text{Inv } X \star X)$
 ⟨*proof*⟩

lemma (in *st-multigroupoid*) *dom-def-var*: $\text{Src } X = \text{sfix} \cap (X \star \text{UNIV})$
 ⟨*proof*⟩

lemma (in *st-multigroupoid*) *cod-def-var*: $\text{Tgt } X = \text{sfix} \cap (\text{UNIV} \star X)$
 ⟨*proof*⟩

lemma (in *st-multigroupoid*) *dom-univ*: $X \star \text{UNIV} = \text{Src } X \star \text{UNIV}$
 ⟨*proof*⟩

lemma (in *st-multigroupoid*) *cod-univ*: $\text{UNIV} \star X = \text{UNIV} \star \text{Tgt } X$
 ⟨*proof*⟩

3.2 Groupoids

Groupoids are simply functional multigroupoids. I start with a somewhat indirect axiomatisation.

class *groupoid-var* = *st-multigroupoid* + *functional-catoid*

begin

lemma *invl* [*simp*]: $x \odot \text{inv } x = \{\sigma x\}$
 ⟨*proof*⟩

lemma *invr* [*simp*]: $\text{inv } x \odot x = \{\tau x\}$
 ⟨*proof*⟩

end

Next, I provide a more direct axiomatisation.

class *groupoid* = *catoid* + *inv-op* +
assumes *invs* [*simp*]: $x \odot \text{inv } x = \{\sigma x\}$

and *inv* [*simp*]: $inv\ x \odot x = \{\tau\ x\}$

subclass (**in** *groupoid*) *st-multigroupoid*
 ⟨*proof*⟩

sublocale *groupoid* \subseteq *lrgpd*: *groupoid* $\lambda x\ y.\ y \odot x$ *tgt src inv*
 ⟨*proof*⟩

lemma (**in** *groupoid*) *ba04* [*simp*]: $inv\ (inv\ x) = x$
 ⟨*proof*⟩

lemma (**in** *groupoid*) *rev1*:
 $x \in y \odot z \implies y \in x \odot inv\ z$
 ⟨*proof*⟩

lemma (**in** *groupoid*) *rev2*:
 $x \in y \odot z \implies z \in inv\ y \odot x$
 ⟨*proof*⟩

lemma (**in** *groupoid*) *rev1-eq*: $(y \in x \odot (inv\ z)) = (x \in y \odot z)$
 ⟨*proof*⟩

lemma (**in** *groupoid*) *rev2-eq*: $(z \in (inv\ y) \odot x) = (x \in y \odot z)$
 ⟨*proof*⟩

The following fact show that the axiomatisation above captures indeed groupoids.

lemma (**in** *groupoid*) *lr-mgpd-partial*:
assumes $x \in y \odot z$
and $x' \in y \odot z$
shows $x = x'$
 ⟨*proof*⟩

subclass (**in** *groupoid*) *single-set-category*
 ⟨*proof*⟩

Hence st-groupoids are indeed single-set categories in which all arrows are isomorphisms.

lemma (**in** *groupoid*) *src-canc1*:
assumes $\tau\ z = \sigma\ x$
and $\tau\ z = \sigma\ y$
and $z \otimes x = z \otimes y$
shows $x = y$
 ⟨*proof*⟩

lemma (**in** *groupoid*) *tgt-canc1*:
assumes $\tau\ x = \sigma\ z$
and $\tau\ y = \sigma\ z$
and $x \otimes z = y \otimes z$

shows $x = y$
<proof>

The following lemmas are from Theorem 5.2 of Jónsson and Tarski's BAO II article.

lemma (in *groupoid*) *ba01* [*simp*]: $x \otimes (\text{inv } x \otimes x) = x$
<proof>

lemma (in *groupoid*) *ba02* [*simp*]: $(x \otimes \text{inv } x) \otimes x = x$
<proof>

lemma (in *groupoid*) *ba05*:
 $\tau x = \sigma y \implies \text{inv } x \otimes x = y \otimes \text{inv } y$
<proof>

lemma (in *groupoid*) *ba06*: $\text{Inv } (x \odot y) = \text{inv } y \odot \text{inv } x$
<proof>

3.3 Axioms of relation algebra

I formalise a special case of a famous theorem of Jónsson and Tarski, showing that groupoids lift to relation algebras at powerset level. All axioms not related to converse have already been considered previously.

lemma (in *groupoid*) *Inv-invol* [*simp*]: $\text{Inv } (\text{Inv } X) = X$
<proof>

lemma (in *groupoid*) *Inv-contrav*: $\text{Inv } (X \star Y) = \text{Inv } Y \star \text{Inv } X$
<proof>

lemma (in *groupoid*) *residuation*: $\text{Inv } X \star -(X \star Y) \subseteq -Y$
<proof>

lemma (in *groupoid*) *modular-law*: $(X \star Y) \cap Z \subseteq (X \cap (Z \star \text{Inv } Y)) \star Y$
<proof>

lemma (in *groupoid*) *dedekind*: $(X \star Y) \cap Z \subseteq (X \cap (Z \star \text{Inv } Y)) \star (Y \cap (\text{Inv } X \star Z))$
<proof>

In sum, this shows that the powerset lifting of a groupoid is a relation algebra. I link this formally with relations in an interpretation statement in another component.

Jónsson and Tarski's axioms of relation algebra are slightly different. It is routine to related them formally with those used here. It might also be interested to use their partiality-by-closure approach to defining groupoids in a setting with explicit carrier sets in another Isabelle formalisation.

lemma (in *groupoid*) *Inv-compl*: $\text{Inv } (-X) = -(\text{Inv } X)$

<proof>

lemma (in *groupoid*) *Inv-inter*: $Inv (X \cap Y) = Inv X \cap Inv Y$
<proof>

lemma (in *groupoid*) *Inv-Un*: $Inv (\bigcap \mathcal{X}) = (\bigcap X \in \mathcal{X}. Inv X)$
<proof>

end

4 Lifting catoids to modal powerset quantales

theory *Catoid-Lifting*

imports *Catoid Quantales-Converse.Modal-Quantale*

begin

instantiation *set* :: (*catoid*) *monoid-mult*

begin

definition *one-set* :: 'a set **where**

$1 = \text{fix}$

definition *times-set* :: 'a set \Rightarrow 'a set \Rightarrow 'a set **where**

$X * Y = X \star Y$

instance

<proof>

end

instantiation *set* :: (*catoid*) *semiring-one-zero*

begin

definition *zero-set* :: 'a set **where**

$\text{zero-set} = \{\}$

definition *plus-set* :: 'a set \Rightarrow 'a set \Rightarrow 'a set **where**

$X + Y = X \cup Y$

instance

<proof>

end

instantiation *set* :: (*catoid*) *dioid*

```

begin

instance
  ⟨proof⟩

end

instantiation set :: (local-catoid) domain-semiring

begin

definition domain-op-set :: 'a set ⇒ 'a set where
  dom X = Src X

instance
  ⟨proof⟩

end

instantiation set :: (local-catoid) range-semiring

begin

definition range-op-set :: 'a set ⇒ 'a set where
  cod X = Tgt X

instance
  ⟨proof⟩

end

instantiation set :: (local-catoid) dr-modal-semiring

begin

instance
  ⟨proof⟩

end

instantiation set :: (catoid) quantale

begin

instance
  ⟨proof⟩

end

```

instantiation *set* :: (*local-catoid*) *domain-quantale*

begin

instance
 <proof>

end

instantiation *set* :: (*local-catoid*) *codomain-quantale*

begin

instance
 <proof>

end

instantiation *set* :: (*local-catoid*) *dc-modal-quantale*

begin

instance
 <proof>

end

end

5 Lifting groupoids to powerset Dedekind quantales and powerset relation algebras

theory *Groupoid-Lifting*

imports *Groupoid Quantales-Converse.Quantale-Converse Catoid-Lifting Relation-Algebra.Relation-Algebra*

begin

instantiation *set* :: (*groupoid*) *dedekind-quantale*

begin

definition *invol-set* :: '*a set* \Rightarrow '*a set* **where**
 invol = Inv

instance
 <proof>

```

end

instantiation set :: (groupoid) boolean-dedekind-quantale

begin

instance⟨proof⟩

end

instantiation set :: (groupoid) relation-algebra

begin

definition composition-set :: 'a set ⇒ 'a set ⇒ 'a set where
  composition-set x y = x ★ y

definition converse-set :: 'a set ⇒ 'a set where
  converse = Inv

definition unit-set :: 'a set where
  unit-set = sfix

instance
  ⟨proof⟩

end

end

```

6 Multimonooids

```

theory Multimonoid
  imports Catoid

```

```

begin

```

```

context multimagma
begin

```

6.1 Unital multimagnas

This component presents an alternative approach to catoids, as multisemigroups with many units. This is more akin to the formalisation of single-set categories in Chapter I of Mac Lane's book, but in fact this approach to axiomatising categories goes back to the middle of the twentieth century.

Units can already be defined in multimagnas.

definition $munitl\ e = ((\exists x. x \in e \odot x) \wedge (\forall x\ y. y \in e \odot x \longrightarrow y = x))$

definition $munitr\ e = ((\exists x. x \in x \odot e) \wedge (\forall x\ y. y \in x \odot e \longrightarrow y = x))$

abbreviation $munit\ e \equiv (munitl\ e \vee munitr\ e)$

end

A multimagma is unital if every element has a left and a right unit.

class $unital-multimagma-var = multimagma +$
 assumes $munitl-ex: \forall x. \exists e. munitl\ e \wedge \Delta\ e\ x$
 assumes $munitr-ex: \forall x. \exists e. munitr\ e \wedge \Delta\ x\ e$

begin

lemma $munitl-ex-var: \forall x. \exists e. munitl\ e \wedge x \in e \odot x$
 $\langle proof \rangle$

lemma $unitl: \bigcup \{e \odot x \mid e. munitl\ e\} = \{x\}$
 $\langle proof \rangle$

lemma $munitr-ex-var: \forall x. \exists e. munitr\ e \wedge x \in x \odot e$
 $\langle proof \rangle$

lemma $unitr: \bigcup \{x \odot e \mid e. munitr\ e\} = \{x\}$
 $\langle proof \rangle$

end

Here is an alternative definition.

class $unital-multimagma = multimagma +$
 fixes $E :: 'a\ set$
 assumes $El: \bigcup \{e \odot x \mid e. e \in E\} = \{x\}$
 and $Er: \bigcup \{x \odot e \mid e. e \in E\} = \{x\}$

begin

lemma $E1: \forall e \in E. (\forall x\ y. y \in e \odot x \longrightarrow y = x)$
 $\langle proof \rangle$

lemma $E2: \forall e \in E. (\forall x\ y. y \in x \odot e \longrightarrow y = x)$
 $\langle proof \rangle$

lemma $El11: \forall x. \exists e \in E. x \in e \odot x$
 $\langle proof \rangle$

lemma $El12: \forall x. \exists e \in E. e \odot x = \{x\}$
 $\langle proof \rangle$

lemma *Er11*: $\forall x. \exists e \in E. x \in x \odot e$
<proof>

lemma *Er12*: $\forall x. \exists e \in E. x \odot e = \{x\}$
<proof>

Units are "orthogonal" idempotents.

lemma *unit-id*: $\forall e \in E. e \in e \odot e$
<proof>

lemma *unit-id-eq*: $\forall e \in E. e \odot e = \{e\}$
<proof>

lemma *unit-comp*:

assumes $e_1 \in E$

and $e_2 \in E$

and $\Delta e_1 e_2$

shows $e_1 = e_2$

<proof>

lemma *unit-comp-iff*: $e_1 \in E \implies e_2 \in E \implies (\Delta e_1 e_2 = (e_1 = e_2))$
<proof>

lemma $\forall e \in E. \exists x. x \in e \odot x$
<proof>

lemma $\forall e \in E. \exists x. x \in x \odot e$
<proof>

sublocale *unital-multimagma-var*
<proof>

Now it is clear that the two definitions are equivalent.

The next two lemmas show that the set of units is a left and right unit of composition at powerset level.

lemma *conv-unl*: $E \star X = X$
<proof>

lemma *conv-unr*: $X \star E = X$
<proof>

end

6.2 Multimonoids

A multimonoid is a unital multisemigroup.

class *multimonoid* = *multisemigroup* + *unital-multimagma*

begin

In a multimonoid, left and right units are unique for each element.

lemma *munits-unique1*: $\forall x. \exists! e. e \in E \wedge e \odot x = \{x\}$
<proof>

lemma *munits-unique2*: $\forall x. \exists! e. e \in E \wedge x \odot e = \{x\}$
<proof>

In a monoid, there is of course one single unit, and my definition of many units reduces to this one.

lemma *units-unique*: $(\forall x y. \Delta x y) \implies \exists! e. e \in E$
<proof>

lemma *units-rm2l*: $e_1 \in E \implies e_2 \in E \implies \Delta e_1 x \implies \Delta e_2 x \implies e_1 = e_2$
<proof>

lemma *units-rm2r*: $e_1 \in E \implies e_2 \in E \implies \Delta x e_1 \implies \Delta x e_2 \implies e_1 = e_2$
<proof>

One can therefore express the functional relationship between elements and their units in terms of explicit (source and target) maps – as in catoids.

definition *so* :: 'a \Rightarrow 'a **where**
so x = (THE e. e \in E \wedge e \odot x = {x})

definition *ta* :: 'a \Rightarrow 'a **where**
ta x = (THE e. e \in E \wedge x \odot e = {x})

abbreviation *So* :: 'a set \Rightarrow 'a set **where**
So X \equiv image *so* X

abbreviation *Ta* :: 'a set \Rightarrow 'a set **where**
Ta X \equiv image *ta* X

end

6.3 Multimonoids and catoids

It is now easy to show that every catoid is a multimonoid and vice versa.

One cannot have both sublocale statements at the same time.

The converse direction requires some preparation.

lemma (in *multimonoid*) *so-unit*: *so* x \in E
<proof>

lemma (in *multimonoid*) *ta-unit*: *ta* x \in E

<proof>

lemma (in *multimonoid*) *so-absorbl*: $so\ x \odot x = \{x\}$
<proof>

lemma (in *multimonoid*) *ta-absorbr*: $x \odot ta\ x = \{x\}$
<proof>

lemma (in *multimonoid*) *semi-locality*: $\Delta\ x\ y \implies ta\ x = so\ y$
<proof>

sublocale *multimonoid* \subseteq *monlr*: *catoid* (\odot) *so ta*
<proof>

6.4 From multimonoids to categories

Single-set categories are precisely local partial monoids, that is, object-free categories as in Chapter I of Mac Lane's book.

class *local-multimagma* = *multimagma* +
assumes *locality*: $v \in x \odot y \implies \Delta\ y\ z \implies \Delta\ v\ z$

class *local-multisemigroup* = *multisemigroup* + *local-multimagma*

In this context, a semicategory is an object-free category without identity arrows

class *of-semicategory* = *local-multisemigroup* + *functional-semigroup*

begin

lemma *part-locality*: $\Delta\ x\ y \implies \Delta\ y\ z \implies \Delta\ (x \otimes y)\ z$
<proof>

lemma *part-locality-var*: $\Delta\ x\ y \implies \Delta\ y\ z \implies (x \odot y) \star \{z\} \neq \{\}$
<proof>

lemma *locality-iff*: $(\Delta\ x\ y \wedge \Delta\ y\ z) = (\Delta\ x\ y \wedge \Delta\ (x \otimes y)\ z)$
<proof>

lemma *locality-iff-var*: $(\Delta\ x\ y \wedge \Delta\ y\ z) = (\Delta\ x\ y \wedge (x \odot y) \star \{z\} \neq \{\})$
<proof>

end

class *partial-monoid* = *multimonoid* + *functional-magma*

class *local-multimonoid* = *multimonoid* + *local-multimagma*

begin

lemma *sota-locality*: $ta\ x = so\ y \implies \Delta\ x\ y$
<proof>

lemma *So-local*: $So\ (x \odot so\ y) = So\ (x \odot y)$
<proof>

lemma *Ta-local*: $Ta\ (ta\ x \odot y) = Ta\ (x \odot y)$
<proof>

sublocale *locmm*: *local-catoid* (\odot) *so ta*
<proof>

The following statements formalise compatibility properties.

lemma *local-conv*: $v \in x \odot y \implies (\Delta\ v\ z = \Delta\ y\ z)$
<proof>

lemma *local-alt*: $e \in E \implies x \in x \odot e \implies y \in e \odot y \implies \Delta\ x\ y$
<proof>

lemma *local-iff*: $\Delta\ x\ y = (\exists e \in E. \Delta\ x\ e \wedge \Delta\ e\ y)$
<proof>

lemma *local-iff2*: $(ta\ x = so\ y) = \Delta\ x\ y$
<proof>

end

Finally I formalise object-free categories. The axioms are essentially Mac Lane's, but a multioperation is used for arrow composition, to capture partiality.

class *of-category* = *of-semicategory* + *partial-monoid*

The next statements show that single-set categories based on catoids and object-free categories based on multimonooids are the same (we can only have one direction as a sublocale statement). It then follows from results about catoids and single-set categories that object-free categories are indeed categories. These results can be found in the catoid component. I do not present explicit proofs for object-free categories here.

sublocale *of-category* \subseteq *ofss-cat*: *single-set-category* - *so ta*
<proof>

6.5 Multimonooids and relational monooids

Relational monooids are monooids in the category Rel. They have been used previously to construct convolution algebras in another AFP entry. Here I show that relational monooids are isomorphic to multimonooids, but I do not

integrate the AFP entry with relational monoids because it uses a historic quantale component, which is different from the quantale component in the AFP. Instead, I simply copy in the definitions leading to relational monoids and leave the consolidation of Isabelle theories to the future.

```

class rel-magma =
  fixes  $\varrho :: 'a \Rightarrow 'a \Rightarrow 'a \Rightarrow bool$ 

class rel-semigroup = rel-magma +
  assumes rel-assoc:  $(\exists y. \varrho y u v \wedge \varrho x y w) = (\exists z. \varrho z v w \wedge \varrho x u z)$ 

class rel-monoid = rel-semigroup +
  fixes  $\xi :: 'a \text{ set}$ 
  assumes unitl-ex:  $\exists e \in \xi. \varrho x e x$ 
  and unitr-ex:  $\exists e \in \xi. \varrho x x e$ 
  and unitl-eq:  $e \in \xi \Longrightarrow \varrho x e y \Longrightarrow x = y$ 
  and unitr-eq:  $e \in \xi \Longrightarrow \varrho x y e \Longrightarrow x = y$ 

```

Once again, only one of the two sublocale statements compiles.

```

sublocale multimonoid  $\subseteq$  rel-monoid  $\lambda x y z. x \in y \odot z E$ 
   $\langle proof \rangle$ 

end

```

References

- [1] R. Brown. From groups to groupoids: A brief survey. *Bulletin of the London Mathematical Society*, 19:113–134, 1987.
- [2] C. Calk, U. Fahrenberg, C. Johansen, G. Struth, and K. Ziemiański. *lr*-multisemigroups, modal quantales and the origin of locality. In *RAMiCS 2021*, volume 13027 of *LNCS*, pages 90–107. Springer, 2021.
- [3] C. Calk, P. Malbos, D. Pous, and G. Struth. Higher catoids, higher quantales and their correspondences. *arXiv*, 2307.09253, 2023.
- [4] J. Cranch, S. Doherty, and G. Struth. Relational semigroups and object-free categories. *arXiv*, 2001.11895, 2020.
- [5] B. Dongol, V. B. F. Gomes, I. J. Hayes, and G. Struth. Partial semigroups and convolution algebras. *Archive of Formal Proofs*, 2017.
- [6] U. Fahrenberg, C. Johansen, G. Struth, and K. Ziemiański. Catoids and modal convolution algebras. *Algebra Universalis*, 84:10, 2023.
- [7] B. Jónsson and A. Tarski. Boolean algebras with operators. Part II. *American Journal of Mathematics*, 74(1):127–162, 1952.

- [8] S. Mac Lane. *Categories for the Working Mathematician*, volume 5. Springer, second edition, 1998.
- [9] E. W. Stark. Category theory with adjunctions and limits. *Archive of Formal Proofs*, 2016.